
Cartouche du document

Année : ING 2
Activité : Travail dirigé

Objectifs
Revoir le langage SQL étudié en première année.

Approfondir la notion du before et after dans un trigger.

Appliquer l'utilisation de curseur

Sommaire des exercices
1 - Les séquences
2 - Impression d'une facture

Corps des exercices
1 - Les séquences
Enoncé :

Soit le script de table suivant :

CREATE TABLE test_table
(idseq number constraint pk_test_table primary key,
col varchar2(10)
);
Le but de cet exercice est de créer une séquence et d'utiliser un trigger en insert sur la table
test_table pour mettre à jour la clé primaire à l'aide de la séquence.

Question 1)

Enoncé de la question

Créer la table à l'aide du script donné dans l'énoncé

Solution de la question

CREATE TABLE test_table
(idseq number constraint pk_test_table primary key,
col varchar2(10)
);

Question 2)

Enoncé de la question

créer une séquence qui s'incrémentera de 1. On appelera cette séquence test_seq

Solution de la question

CREATE SEQUENCE test_seq START WITH 1 INCREMENT BY 1;

Question 3)

Enoncé de la question

utiliser la séquence pour insérer une ligne dans la table test_table

Solution de la question

insert into test_table values( test_seq.nextval,'ligne1');

Question 4)

Enoncé de la question

Ecrire en PL/SQL un déclencheur (trigger) qui permettra de mettre à jour la clé primaire en utilisant
la séquence.

Solution de la question

Le trigger doit obligatoirement être crée before insert.

create or replace trigger trig_test_table_pk
before insert on test_table for each row
begin

select test_seq.nextval into :new.idseq from dual;
end;
/

Pour tester le trigger :

insert into test_table values(null,'ligne2')

2 - Impression d'une facture
Enoncé :

Reprendre l'exercice 2 du tp précédent

On désire maintenant afficher la facture correpondant à une commande. Sur cette facture doit
apparaître : le client,le numéro de la commande et la date de la commande, le montant total ainsi
que pour chaque produit commandé, la quantité, le prix et le montant à payer correspondant.

Question 1)

Enoncé de la question

Ecrire la procédure qui acceptera en paramètre le numéro de la commande et qui permettra
d'afficher la facture.

Solution de la question

create or replace procedure facture(pidcommande varchar2)
is
-- déclaration des variables utiles
vidcommande commande.idcommande%type ;
vidclient commande.idclient%type ;
vdatecde commande.datecde%type ;
vmontanttot commande.montanttotalcom%type ;
vmontant commande.montanttotalcom%type ;

-- déclaration du curseur qui contiendra les lignes de commande
-- pour une commande (vidcommande) concernée

cursor cur_lignecde(vidcommande varchar2) is
select idlig,quantiteprod,idproduit
from lignecommande
where idcommande = vidcommande ;

begin
-- on initialise la variable vidcommande avec la valeur passée en paramètre

vidcommande :=pidcommande ;

-- on récupère les valeurs de la table commande pour la commande -- --concernée(pidcommande)

select idclient,datecde,montanttotalcom into vidclient,vdatecde,vmontanttot
from commande
where idcommande = vidcommande ;

-- on affiche l entête de la facture

dbms_output.put_line('FACTURE client numéro ' || vidclient) ;
dbms_output.put_line('Commande numéro : '|| vidcommande);
dbms_output.put_line('Date de la commande : '|| vdatecde);
dbms_output.put_line('---------------------------------------------') ;
dbms_output.put_line('') ;

-- on gère les lines de commande à l aide du curseur

for vcur in cur_lignecde(pidcommande)
loop

-- pour chaque ligne on récupère le prix et on calcul le montan pour le produit
-- commandé

select prixprod*vcur.quantiteprod into vmontant from produit where idproduit=vcur.idproduit ;

-- on affiche les info demandés pour chaque produit commandé

dbms_output.put_line
(vcur.idlig || ' ' ||vcur.idproduit||' '||vcur.quantiteprod||' '||vmontant);

end loop ;

-- on affiche le pied de la facture
dbms_output.put_line('---------------------------------------------') ;
dbms_output.put_line('Montant total de la commande : ' || vmontanttot );
exception
when no_data_found then
dbms_output.put_line('erreur dans le numéro de la commande') ;
end ;
/

